

SCIENCE & VIE

MENSUEL N° 883

AVRIL 1991

LE PLAISIR QUI TUE (5)
DROGUE:

- Des vaches folles en France !

- Intelligence artificielle : une découverte



**"MHD":
LA VITESSE
SANS MOTEUR
ET SANS HÉLICE**

M 2578 - 883 - 20,00 F



145 FB - 6 FS - Scan 3.75 - 550 Ptas - 550 Esc - 28 Dh - 2.750 Dt -
4800 L - USA NYC \$ 4.25 - RCI: 1370 CFA - 940 CFP

Un jeu de taquin

INFORMATIQUE AMUSANTE

Qui n'a jamais eu entre les mains un jeu de taquin ? Cette petite plaquette sur laquelle coulisent, grâce à la présence d'un emplacement libre, des carrés portant lettres et chiffres, permet de former des mots ou des nombres en déplaçant les carrés de base. Ancêtre du Rubick's cube, bien que nettement moins compliqué, ce jeu a également tendance à vous chatouiller les méninges, surtout dès que le mot à composer devient un tant soit peu long. Par contre, il est facile de doser le niveau de difficulté tout simplement par le choix de la longueur du mot à afficher. Il est donc possible,

pour débiter, de commencer par des mots de trois lettres par exemple puis, progressivement, de s'attaquer à des mots de plus en plus longs.

Précisons, toutefois, que ce jeu possède une limite incontournable en ce qui concerne le choix des mots. En effet chaque lettre n'est disponible qu'une seule fois sur la grille. Donc, notamment pour les mots longs, ce choix restera relativement limité.

Dans notre cas, nous disposons de l'alphabet complet et des chiffres allant de 1 à 9. Soit au total 35 cases de base plus l'espace vide, soit encore une grille comportant 6

rangs de 6 cases chacune.

Ces quelques précisions apportées, passons à l'écriture de notre programme. En fait, pour l'ordinateur, le plus important reste de se "rappeler" la position de chaque carré de base sur la grille à tout instant. Pour cela, nous commencerons par créer un tableau de mémorisation comportant autant de cases qu'il y a de carrés pouvant être déplacés, y compris la case vide.

Il comportera donc 36 cases réparties 6 par 6, disposition au total identique à celle de la grille de jeu, afin de retrouver plus rapidement l'emplacement souhaité dès que la machine devra venir y piocher les informations nécessaires au déplacement d'une case.

Nous redéfinirons ensuite les symboles graphiques nécessaires à la présentation du jeu sur l'écran et la grille sera affichée "rangée" : les

```

10 REM *****
20 REM *
30 REM * AFFICHAGE DES COMMENTAIRES. *
40 REM *
50 REM *****
60 MODE 1:CLS
70 LOCATE 18,1:PRINT "JEU DE TAQUIN."
80 LOCATE 18,5:PRINT "UTILISEZ LES TOUCHES"
90 LOCATE 18,7:PRINT "MARQUEES D'UNE FLECHE"
100 LOCATE 18,9:PRINT "POUR AMENER LES CARRES"
110 LOCATE 18,11:PRINT "SUR L'EMPLACEMENT"
120 LOCATE 18,13:PRINT "VIDE."
130 LOCATE 1,17:PRINT "EN DEBUT DE PARTIE: "
140 LOCATE 1,19:PRINT "LA FLECHE VERS LA DROITE Y
PLACERA LE 9"
150 LOCATE 1,21:PRINT "LA FLECHE VERS LE BAS Y PLA
CERA LE 4."
160 REM *****
170 REM *
180 REM * INITIALISATION DES VARIABLES ET *
190 REM * CREATION DU TABLEAU DE MEMORISATION. *
200 REM *
210 REM *****
220 LET XV=6:LET YV=6
230 DIM C$(6,6)
240 REM *****
250 REM *
260 REM * REDEFINITION DES SYMBOLES GRAPHIQUES *
270 REM *
280 REM *****
290 SYMBOL AFTER 127
300 SYMBOL 128,24,24,60,255,255,60,24,24
310 SYMBOL 129,24,24,24,24,24,24,24,24
320 SYMBOL 130,0,0,0,255,255,0,0,0
330 SYMBOL 131,240,240,240,240,240,240,240,240
340 SYMBOL 132,255,255,255,255,0,0,0,0
350 SYMBOL 133,15,15,15,15,15,15,15,15
360 SYMBOL 134,0,0,0,0,255,255,255,255
370 SYMBOL 135,255,255,255,255,248,240,240,240
380 SYMBOL 136,255,255,255,255,31,15,15,15
390 SYMBOL 137,15,15,15,31,255,255,255,255
400 SYMBOL 138,240,240,240,240,248,255,255,255
410 SYMBOL 139,255,255,255,255,60,24,24,24
420 SYMBOL 140,24,24,24,60,255,255,255,255
430 SYMBOL 141,15,15,31,255,255;31,15,15

```

```

440 SYMBOL 142,240,240,248,255,255,248,240,240
450 REM *****
460 REM *
470 REM * DONNEES DU CONTENU DE LA GRILLE. *
480 REM *
490 REM *****
500 DATA "A","B","C","D","E","F"
510 DATA "G","H","I","J","K","L"
520 DATA "M","N","O","P","Q","R"
530 DATA "S","T","U","V","W","X"
540 DATA "Y","Z","1","2","3","4"
550 DATA "5","6","7","8","9"," "
560 REM *****
570 REM *
580 REM * REMPLISSAGE DE LA MEMOIRE GRILLE. *
590 REM *
600 REM *****
610 RESTORE 500
620 FOR Y=1 TO 6
630 FOR X=1 TO 6
640 READ L$
650 LET C$(X,Y)=L$
660 NEXT X
670 NEXT Y
680 REM *****
690 REM *
700 REM * AFFICHAGE DE LA GRILLE. *
710 REM *
720 REM *****
730 LOCATE 1,1:PRINT CHR$(135)
740 LOCATE 13,1:PRINT CHR$(136)
750 LOCATE 13,13:PRINT CHR$(137)
760 LOCATE 1,13:PRINT CHR$(138)
770 FOR I= 2 TO 12 STEP 2
780 LOCATE I,1:PRINT CHR$(132)
790 LOCATE I,13:PRINT CHR$(134)
800 LOCATE 1,I:PRINT CHR$(131)
810 LOCATE 13,I:PRINT CHR$(133)
820 NEXT I
830 FOR I= 3 TO 11 STEP 2
840 LOCATE I,1:PRINT CHR$(139)
850 LOCATE I,13:PRINT CHR$(140)
860 LOCATE 1,I:PRINT CHR$(142)
870 LOCATE 13,I:PRINT CHR$(141)
880 NEXT I

```

lettres seront classées par ordre alphabétique et les chiffres par ordre croissant. Quant à l'emplacement libre, il sera placé en bas et à droite de la grille de jeu. Ainsi initialisé, notre tableau sera prêt à évoluer en fonction des coups joués. Notons par ailleurs qu'afin de pouvoir les exécuter nous initialiserons deux variables de déplacement XV et YV. Elle indiqueront à l'ordinateur l'emplacement occupé par la case vide au sein de la grille de jeu.

Une sous-routine sera ensuite utilisée pour prendre en compte les choix du joueur. Cette dernière tournera essentiellement autour de la variable K\$ utilisée pour prendre en compte les touches frappées au clavier. Il ne s'agira ici d'ailleurs que de l'une des quatre flèches de déplacement du curseur. Après la frappe de chacune d'entre elles une série de contrôles sera effectuée par le programme afin de vé-

rifier si le déplacement demandé est effectivement possible. N'oublions pas que l'emplacement libre sera comblé par l'un des carrés adjacents dont la direction du déplacement sera indiquée par la flèche choisie. Si on frappe "flèche vers la gauche", la case vide sera remplacée par le carré placé à sa droite. Il y a donc des cas pour lesquels certaines directions sont interdites, notamment dès que la case vide se trouvera le long d'un bord de la grille ou, *a fortiori*, dans un coin. Dans ces deux cas, si une direction "impossible" est demandée, l'ordinateur ignorera l'ordre de déplacement. La direction des flèches indique le déplacement des carrés à amener dans la case vide parce que ce choix nous a paru logique. Cependant, elle peut sembler inversée pour certains joueurs. Rien ne s'oppose à modifier ce fonctionnement en jouant sur les paramè-

tres de la sous-routine de déplacement.

Un "compteur de coups" indiquera en permanence le nombre de déplacements effectués depuis le début du jeu. La frappe de ce programme ne doit pas poser de problème particulier. Seules quelques instructions particulières au Basic Amstrad devront éventuellement être modifiées pour adapter ce programme à d'autres appareils, mais nous avons déjà indiqué de nombreuses fois dans ces lignes leurs équivalences.

L'utilisation de ce jeu est des plus simples. Une fois RUN demandé, la grille de jeu sera affichée sur l'écran. Chaque fois qu'une touche de déplacement curseur sera frappée, le déplacement des cases sera effectué. Cela sera répété autant de fois qu'il sera nécessaire pour atteindre le but recherché.

Henri-Pierre Penel ▲

```

890 FOR V=3 TO 11 STEP 2
900 FOR H=3 TO 11 STEP 2
910 LOCATE H,V:PRINT CHR$(128)
920 NEXT H
930 NEXT V
940 FOR V=2 TO 12 STEP 2
950 FOR H=3 TO 11 STEP 2
960 LOCATE H,V:PRINT CHR$(129)
970 NEXT H
980 NEXT V
990 FOR V=3 TO 11 STEP 2
1000 FOR H=2 TO 12 STEP 2
1010 LOCATE H,V:PRINT CHR$(130)
1020 NEXT H
1030 NEXT V
1040 GOSUB 1720
1050 REM *****
1060 REM *
1070 REM *      PRISE EN COMPTE DU CHOIX      *
1080 REM *      DU DEPLACEMENT.                *
1090 REM *
1100 REM *****
1110 LET K$=INKEY$:IF K$="" THEN GOTO 1110
1120 LET K=ASC(K$):IF K<240 OR K>243 THEN GOTO 1110
1130 IF K=242 AND XV<6 THEN GOSUB 1240
1140 IF K=243 AND XV>1 THEN GOSUB 1360
1150 IF K=240 AND YV<6 THEN GOSUB 1480
1160 IF K=241 AND YV>1 THEN GOSUB 1600
1170 GOTO 1110
1180 REM *****
1190 REM *
1200 REM *      SOUS ROUTINE DE DEPLACEMENT      *
1210 REM *      D'UN CARRE VERS LA DROITE.        *
1220 REM *
1230 REM *****
1240 LET C$(XV,YV)=C$(XV+1,YV)
1250 LOCATE 2*XV,2*YV:PRINT C$(XV,YV)
1260 LET XV=XV+1
1270 LOCATE 2*XV,2*YV:PRINT " "
1280 LET C$(XV,YV)=" "
1290 RETURN
1300 REM *****
1310 REM *
1320 REM *      SOUS ROUTINE DE DEPLACEMENT      *
1330 REM *      D'UN CARRE VERS LA GAUCHE.        *
1340 REM *
1350 REM *****
1360 LET C$(XV,YV)=C$(XV-1,YV)
1370 LOCATE 2*XV,2*YV:PRINT C$(XV,YV)
1380 LET XV=XV-1
1390 LOCATE 2*XV,2*YV:PRINT " "
1400 LET C$(XV,YV)=" "
1410 RETURN
1420 REM *****
1430 REM *
1440 REM *      SOUS ROUTINE DE DEPLACEMENT      *
1450 REM *      D'UN CARRE VERS LE BAS.          *
1460 REM *
1470 REM *****
1480 LET C$(XV,YV)=C$(XV,YV+1)
1490 LOCATE 2*XV,2*YV:PRINT C$(XV,YV)
1500 LET YV=YV+1
1510 LOCATE 2*XV,2*YV:PRINT " "
1520 LET C$(XV,YV)=" "
1530 RETURN
1540 REM *****
1550 REM *
1560 REM *      SOUS ROUTINE DE DEPLACEMENT      *
1570 REM *      D'UN CARRE VERS LE HAUT.        *
1580 REM *
1590 REM *****
1600 LET C$(XV,YV)=C$(XV,YV-1)
1610 LOCATE 2*XV,2*YV:PRINT C$(XV,YV)
1620 LET YV=YV-1
1630 LOCATE 2*XV,2*YV:PRINT " "
1640 LET C$(XV,YV)=" "
1650 RETURN
1660 REM *****
1670 REM *
1680 REM *      SOUS ROUTINE D'AFFICHAGE DES      *
1690 REM *      CARRÉS EN DEBUT DE JEU.          *
1700 REM *
1710 REM *****
1720 FOR Y=1 TO 6
1730 FOR X=1 TO 6
1740 LOCATE 2*X,2*Y:PRINT C$(X,Y)
1750 NEXT X
1760 NEXT Y
1770 RETURN

```